Regular Paper

高階微分による陰関数定理の形式化について

On The Formalizations for Implicit Function Theorem with Higher Derivatives of Vector Valued Functions

中正 和久 [1,*] 師玉 康成 [2]
Kazuhisa Nakasho[1,*] Yasunari Shidama[2]

1 山口大学大学院創成科学研究科, 山口県宇部市常盤台 2-16-1

1 Graduated School of Science and Technology for Innovation, Yamaguchi University, 2-16-1 Tokiwa-dai, Ube, Japan

2 長野県軽井沢町

2 Karuizawa, Nagano, Japan

* nakasho@yamaguchi-u.ac.jp

Proof checked by Mizar Version: 8.1.12 and MML Version: 5.73.1435
Received: December 6, 2022. Accepted: March 14, 2023.

Abstract

In this paper, we report on the formalization of the implicit function theorem and the inverse function theorem with higher-order derivatives and partial derivatives of vector-valued functions on normed spaces.

1 はじめに

筆者らはこれまでに Laurent Schwartz による著書 [1] にしたがって，[2] でノルム空間上の陰関数定理を，[3] で逆関数定理をそれぞれ以下のように形式化した．

Listing 1. NDIFF_9:21

```
theorem :: NDIFF_9:21
  for E be RealNormSpace,
      G,F be non trivial RealBanachSpace,
      Z be Subset of [:E,F:],
      f be PartFunc of [:E,F:], G,
      a be Point of E, b be Point of F, c be Point of G,
      z be Point of [:E,F:]
  st Z is open & dom f = Z
```

```
  & f is_differentiable_on Z & f `| Z is_continuous_on Z
  & [a,b] in Z & f.(a,b) = c & z = [a,b]
  & partdiff`2(f,z) is invertible
holds
  ex r1,r2 be Real
  st 0 < r1 & 0 < r2
   & [:Ball(a,r1),cl_Ball(b,r2):] c= Z
   & ( for x be Point of E st x in Ball(a,r1) holds
       ex y be Point of F st y in Ball(b,r2) & f.(x,y) = c )
   & ( for x be Point of E st x in Ball(a,r1) holds
       for y1,y2 be Point of F st y1 in Ball(b,r2) & y2 in Ball(b,r2)
        & f.(x,y1) = c & f.(x,y2) = c holds y1=y2 )
   & ( ex g be PartFunc of E,F
       st dom g = Ball(a,r1) & rng g c= Ball(b,r2)
        & g is_continuous_on Ball(a,r1) & g.a = b
        & ( for x be Point of E st x in Ball(a,r1) holds f.(x,g.x) = c )
        & g is_differentiable_on Ball(a,r1)
        & g `| Ball(a,r1) is_continuous_on Ball(a,r1)
        & ( for x be Point of E, z be Point of [:E,F:]
            st x in Ball(a,r1) & z = [x,g.x]
            holds diff(g,x) = - ( Inv partdiff`2(f,z)) * partdiff`1(f,z) )
        & ( for x be Point of E, z be Point of [:E,F:]
            st x in Ball(a,r1) & z =[x,g.x]
            holds partdiff`2(f,z) is invertible ) )
   & ( for g1,g2 be PartFunc of E,F
       st dom g1 = Ball(a,r1) & rng g1 c= Ball(b,r2)
        & ( for x be Point of E st x in Ball(a,r1) holds f.(x,g1.x) = c )
        & dom g2 = Ball(a,r1) & rng g2 c= Ball(b,r2)
        & ( for x be Point of E st x in Ball(a,r1) holds f.(x,g2.x) = c )
            holds g1 = g2);
```

Listing 2. NDIFF10:17

```
theorem :: NDIFF10:17
  for E,F be non trivial RealBanachSpace,
      Z be Subset of E,
      f be PartFunc of E,F,
      a be Point of E, b be Point of F
  st Z is open & dom f = Z
   & f is_differentiable_on Z & f `| Z is_continuous_on Z
   & a in Z & f.a = b
   & diff(f,a) is invertible
  holds
    ex A be Subset of E,B be Subset of F,
       g be PartFunc of F,E
    st A is open & B is open
     & A c= dom f & a in A & b in B
     & f.:A = B & dom g = B & rng g = A
     & dom(f|A) = A & rng(f|A) = B
     & (f|A) is one-to-one & g is one-to-one
     & g = (f|A)" & (f|A) = g" & g.b = a
     & g is_continuous_on B & g is_differentiable_on B
     & g `| B is_continuous_on B & ( for y be Point of F st y in B
         holds diff(f,g/.y) is invertible )
     & ( for y be Point of F st y in B holds diff(g,y) = Inv diff(f,g/.y) );
```

さらに [4] では有限次元空間上でのそれらの行列表現を形式化した．また， [5] で報告した NDIFF12.abs ではノルム空間上での高階微分，偏微分の形式化を行っている．本稿では，これらを用いたノルム空間上の高階連続微分可能関数に関する陰関数定理と逆関数定理の形式化について報告する．

筆者らは，上記の NDIFF_9:21 で存在が保証されている陰関数 g について，以下の NDIFF13:65 のように f が高階連続微分可能であるときに，g も同様に高階連続微分可能であることを示す定理とその証明の形式化を行った．

Listing 3. NDIFF13:65

```
theorem :: NDIFF13:65
  for E be RealBanachSpace, F,G be non trivial RealBanachSpace,
      Z be non empty Subset of [:E,F:],c be Point of G,
      A be Subset of E, B be Subset of F
    st Z is open & A is open & B is open & [:A,B:] c= Z
  holds
    for i be Nat,
        f be PartFunc of [:E,F:],G, g be PartFunc of E,F
     st dom f = Z & f is_differentiable_on i+1,Z
      & diff(f,i+1,Z) is_continuous_on Z
      & dom g = A & rng g c= B & g is_continuous_on A
      & ( for x be Point of E st x in A holds f.(x,g.x) = c )
      & ( for x be Point of E, z be Point of [:E,F:] st x in A & z = [x,g.x]
          holds partdiff`2(f,z) is invertible )
    holds
        g is_differentiable_on i+1,A & diff(g,i+1,A) is_continuous_on A
      & for x be Point of E, z be Point of [:E,F:] st x in A & z = [x,g.x]
        holds diff(g,x) = - (Inv partdiff`2(f,z)) * partdiff`1(f,z);
```

その概要は次の通りである．

E をノルム空間,F, G をバナハ空間 (RealBanachSpace)，Z を E, F の直積空間 $E \times F$ の開部分集合，c を G の点，A, B をそれぞれ E, F の開集合で

$$A \times B \subseteq Z$$

とする．

$$f : (x, y) \in Z \subseteq E \times F \mapsto f(x, y) \in G$$

を Z から G への写像で Z 上で $i+1$ 階連続微分可能とする．また，g を A から B への写像で A に含まれる E の任意の点 $x \in A$ について

$$f(x, g(x)) = c$$

かつ

$$\frac{\partial f}{\partial y}(x, g(x))$$

が可逆であるとする．このとき，g は $i+1$ 階連続微分可能であり，A に含まれる E の任意の点 $x \in A$ について

$$g'(x) = \left(\frac{\partial f}{\partial y}(x, g(x))\right)^{-1} \circ \frac{\partial f}{\partial x}(x, g(x)) \tag{1}$$

である．

さらに，以下の NDIFF13:67 のように，上記の逆関数定理 NDIFF10:17 の関数 f が $n+1$ 階連続微分可能であるとき，その存在が保証される関数 g も $n+1$ 階連続微分可能であることを示す定理とその証明の形式化を行った．

Listing 4. NDIFF13:67

```
theorem :: NDIFF13:67
  for E,F be non trivial RealBanachSpace,
      Z be Subset of E, f be PartFunc of E,F,
      a be Point of E, b be Point of F, n be Nat
  st Z is open & dom f = Z & f is_differentiable_on n+1,Z & diff(f,n+1,Z) is_continuous_on Z
```

```
& a in Z & f.a = b & diff(f,a) is invertible
holds
 ex A be Subset of E,B be Subset of F, g be PartFunc of F,E
 st A is open & B is open & A c= dom f
  & a in A & b in B & f.:A = B
  & dom g = B & rng g = A & dom(f|A) = A & rng(f|A) = B
  & (f|A) is one-to-one & g is one-to-one
  & g = (f|A)" & (f|A) = g" & g.b = a
  & ( for y be Point of F st y in B holds diff(f,g/.y) is invertible )
  & ( for y be Point of F st y in B holds diff(g,y) = Inv diff(f,g/.y) )
  & f is_differentiable_on n+1,A & diff(f,n+1,A) is_continuous_on A
  & g is_differentiable_on n+1,B & diff(g,n+1,B) is_continuous_on B;
```

その概要は次の通りである．

E, F をバナハ空間 (RealBanachSpace)，Z を E の開部分集合，

$$f : x \in Z \subseteq E \mapsto f(x) \in F$$

を Z から F への写像で Z 上で $n+1$ 階連続微分可能とする．a, b をそれぞれ E, F の元とし，a は Z に含まれ

$$f(a) = b \tag{2}$$

$$f'(a) \text{ が可逆} \tag{3}$$

とする．このとき，f の定義域 Z に含まれる a の開近傍 A と b の開近傍 B 及び B から A の上への写像 g が存在し，

$$f(A) = B,\ g(b) = a$$

を満たす．f の A 上への制限 $f|A$ と g は全単射であり，互いに逆写像になっている．しかも，$f|A$ と g はそれぞれ A, B 上 $n+1$ 階連続微分可能である．また，B の任意の点 $y \in B$ について

$$f'(g(y))$$

は可逆であり，

$$g'(y) = f'(g(y))^{-1}$$

を満たす．

2 積の微分

ここで，3 章と 4 章で述べる定理の証明で用いられる中心的な命題について述べておく．

2.1 ライプニッツの公式

ノルム空間 S, E, F について S から E への写像 u と，S から F への写像 v が与えられたとき，これらの写像と S から $E \times F$ への写像

$$u \odot v :\ x \in S \mapsto (u(x), v(x)) \in E \times F \tag{4}$$

が定義できる．

ノルム空間 S, E, F, G と S の開集合 Z，任意の自然数 n について，$E \times F$ から G への有界連続双線形写像

$$B: \ (x, y) \in E \times F \mapsto B(x, y) \in G$$

および，Z 上で n 階連続微分可能写像であるそれぞれ S から E への写像 u と，S から F への写像 v が与えられたとき，これらの写像と S から (4) 式により $E \times F$ への写像

$$u \odot v: \ x \in S \mapsto (u(x), v(x)) \in E \times F$$

を用いて合成される S から F への写像

$$B \circ (u \odot v): \ x \in S \mapsto (B \circ (u \odot v))(x) = B(u(x), v(x)) \in F$$

は n 階連続微分可能である．筆者らはこれを以下のように形式化した．

Listing 5. NDIFF13:42

```
theorem :: NDIFF13:42
  for i be Nat,S,E,F,G be RealNormSpace, Z be Subset of S,
      B be Lipschitzian BilinearOperator of E,F,G,
      u be PartFunc of S,E, v be PartFunc of S,F,
      w be PartFunc of S,[:E,F:], W be PartFunc of S,G
   st W = B * w & w = <:u,v:>
    & u is_differentiable_on i,Z & diff(u,i,Z) is_continuous_on Z
    & v is_differentiable_on i,Z & diff(v,i,Z) is_continuous_on Z
  holds
    W is_differentiable_on i,Z & diff(W,i,Z) is_continuous_on Z;
```

この命題の証明には微分の階数 n についての帰納法を用いる．

$n = 1$ についてはまず，

Listing 6. NDIFF12:14

```
theorem :: NDIFF12:14
  for f be Lipschitzian BilinearOperator of E,F,G, Z be Subset of [:E,F:]
   st Z is open
  holds
    f is_differentiable_on Z & f `| Z is_continuous_on Z;
```

から有界連続双線形写像 B はその定義域 $E \times F$ 上連続微分可能である．

また写像 $u \odot v$ は

Listing 7. NDIFF13:29

```
theorem :: NDIFF13:29
  for S,T,U be RealNormSpace,Z be Subset of S,
      u be PartFunc of S,T, v be PartFunc of S,U,
      w be PartFunc of S,[:T,U:]
    st u is_differentiable_on Z & u `| Z is_continuous_on Z
     & v is_differentiable_on Z & v `| Z is_continuous_on Z
     & w = <:u,v:>
  holds
     w is_differentiable_on Z & w `| Z is_continuous_on Z;
```

から Z 上連続微分可能である．

写像 B と $u \odot v$ の合成 $B \circ (u \odot v)$ は

Listing 8. NDIFF13:23

```
theorem :: NDIFF13:23
  for E,F,G be RealNormSpace, Z be Subset of E,
      T be Subset of F, u be PartFunc of E,F,
      v be PartFunc of F,G
   st u .: Z c= T
    & u is_differentiable_on Z & u `| Z is_continuous_on Z
    & v is_differentiable_on T & v `| T is_continuous_on T
  holds
     v*u is_differentiable_on Z & v*u `| Z is_continuous_on Z;
```

から Z 上連続微分可能である．

次に帰納法の仮定により微分の階数 n について定理が成り立つものとする．以後の議論はライプニッツの公式

$$(B \circ (u \odot v))'(x)\,ds = B(u'(x)\,ds, v(x)) + B(u(x), v'(x)\,ds) \tag{5}$$

による．これは以下のように形式化されている．

Listing 9. NDIFF13:35

```
theorem :: NDIFF13:35
  for S,E,F,G be RealNormSpace,Z be Subset of S,
      B be Lipschitzian BilinearOperator of E,F,G,
      W be PartFunc of S,G, w be PartFunc of S,[:E,F:],
      u be PartFunc of S,E, v be PartFunc of S,F
   st u is_differentiable_on Z & v is_differentiable_on Z
    & W = B * w & w = <:u,v:>
  holds
    W is_differentiable_on Z
  & for x be Point of S st x in Z holds
      for ds be Point of S
      holds
        ((W`|Z)/.x).ds = B.(((u`|Z)/.x).ds, v/.x) + B.(u/.x, ((v`|Z)/.x).ds);
```

(5) 式の左辺はノルム空間 S の元 ds に S から G への有界線形写像が作る空間 $\mathcal{L}(S,G)$ の元である有界線形写像

$$(B \circ (u \odot v))'(x)$$

が作用していることを表している．

$$(B \circ (u \odot v))'$$

は S から $\mathcal{L}(S,G)$ への写像である．

F の元 $v \in F$ に関して写像

$$B_v : z \in E \mapsto B(z,v)$$

は E から G への有界線形写像が作る空間 $\mathcal{L}(E,G)$ の元である．

写像

$$B_1 : v \in F \mapsto B_v \in \mathcal{L}(E,G)$$

は F から $\mathcal{L}(E,G)$ への有界線形写像である．

(5) 式の右辺第 1 項は

$$\begin{aligned} B(u'(x)\,ds, v(x)) &= B_{v(x)}(u'(x)\,ds) \\ &= (B_1 \circ v)(x)(u'(x)\,ds) \\ &= ((B_1 \circ v)(x) \circ u'(x))\,ds \end{aligned} \quad (6)$$

と書き表せる．同様に E の元 $u \in E$ に関して写像

$$B_u : z \in F \mapsto B(u, z)$$

は F から G への有界線形写像が作る空間 $\mathcal{L}(F, G)$ の元である．

写像

$$B_2 : u \in E \mapsto B_u \in \mathcal{L}(F, G)$$

は E から $\mathcal{L}(F, G)$ への有界線形写像である．

(5) 式の右辺第 2 項は

$$B(u(x), v'(x)\,ds) = ((B_2 \circ u)(x) \circ v'(x))\,ds \quad (7)$$

と書き表せる．

(6)(7) 式から，(5) 式は

$$((B \circ (u \odot v))'(x))\,ds = ((B_1 \circ v)(x) \circ u'(x))\,ds + ((B_2 \circ u)(x) \circ v'(x))\,ds \quad (8)$$

と書き表せる．これから

$$(B \circ (u \odot v))'(x) = (B_1 \circ v)(x) \circ u'(x) + (B_2 \circ u)(x) \circ v'(x) \quad (9)$$

を得る．

写像 $B_1 \circ v$ は有界線形写像 B_1 と n 階連続微分可能な写像 v との合成であり，以下の命題により n 階連続微分可能である．

Listing 10. NDIFF13:26

```
theorem :: NDIFF13:26
  for E,F,G be RealNormSpace, Z be Subset of E,
      u be PartFunc of E,F,L be Lipschitzian LinearOperator of F,G
  holds
    for i be Nat
     st u is_differentiable_on i,Z & diff(u,i,Z) is_continuous_on Z
    holds L*u is_differentiable_on i,Z & diff(L*u,i,Z) is_continuous_on Z;
```

また，仮定から u は $n+1$ 階連続微分可能であるため，du/dx は n 階連続微分可能である．写像の合成は有界双線形写像であるから，帰納法の仮定から

$$(B_1 \circ v)(x) \circ u'(x)$$

は n 階連続微分可能である．同様に，

$$(B_2 \circ u)(x) \circ v'(x)$$

も n 階連続微分可能である．

以上から (9) 式により

$$(B \circ (u \odot v))'$$

は n 階連続微分可能である．

最後に

Listing 11. NDIFF13:33

```
theorem :: NDIFF13:33
  for E,F be RealNormSpace,
      n be Nat, Z be Subset of E,
      g be PartFunc of E,F
   st g`|Z is_differentiable_on n,Z & g is_differentiable_on Z
    & diff(g`|Z,n,Z) is_continuous_on Z
  holds g is_differentiable_on n+1,Z & diff(g,n+1,Z) is_continuous_on Z;
```

を適用して，$B \circ (u \odot v)$ が $n+1$ 階連続微分可能であることが示される．以上で微分の階数に関する帰納法の証明が完成する．

2.2　合成関数の微分

前項の命題を用いれば以下のように 2 つの連続微分可能な写像の合成写像が連続微分可能になることを示すことができる．

ノルム空間 E, F, G と部分関数 $u : E \to F$，$v : F \to G$ に対して，u は $Z(\subset E)$ 上で，v は $T(\subset F)$ 上でそれぞれ n 階連続微分可能で，

$$u(Z) \subseteq T$$

を満たすものとする．このとき，u, v の合成写像

$$v \circ u$$

は，Z 上で n 階連続微分可能写像である．これを形式化したものが以下である．

Listing 12. NDIFF13:44

```
theorem :: NDIFF13:44
  for i be Nat, E,F,G be RealNormSpace, Z be Subset of E,
      T be Subset of F, u be PartFunc of E,F,
      v be PartFunc of F,G
   st u.:Z c= T
    & u is_differentiable_on i,Z & diff(u,i,Z) is_continuous_on Z
    & v is_differentiable_on i,T & diff(v,i,T) is_continuous_on T
  holds
      v*u is_differentiable_on i,Z & diff(v*u,i,Z) is_continuous_on Z;
```

この命題の証明には微分の階数 n についての帰納法を用いる．

$n = 1$ については NDIFF13:23 で証明されている．微分の階数が n の場合に成り立つものとする．u, v が $n+1$ 階連続微分可能であるとする．微分の階数を下げるために，合成関数の微分

$$(v \circ u)'(x) = v'(u(x)) \circ u'(x) = (v' \circ u)(x) \circ u'(x) \tag{10}$$

を用いる．これは以下で形式化されている．

Listing 13. NDIFF13:19

```
theorem :: NDIFF13:19
  for E,F,G be RealNormSpace, Z be Subset of E,
      T be Subset of F, u be PartFunc of E,F, v be PartFunc of F,G
   st u.:Z c= T
    & u is_differentiable_on Z & v is_differentiable_on T
  holds
    v*u is_differentiable_on Z
  & for x be Point of E st x in Z
    holds ((v*u) `| Z) /. x = ((v`| T) /. (u /. x)) * ((u`| Z)/.x);
```

u, v が $n+1$ 階連続微分可能であるから，u は n 階連続微分可能であり，v' も n 階連続微分可能である．したがって，帰納法の仮定から，(10) 式中の関数 $v' \circ u$ も n 階連続微分可能である．これは E からノルム空間 F から G への有界線形写像が作る空間 $\mathcal{L}(F, G)$ への写像である．

u' は，E からノルム空間 E から F への有界線形写像が作る空間 $\mathcal{L}(E, F)$ への n 階連続微分可能な写像である．ここで，それぞれ $\mathcal{L}(E, F)$，$\mathcal{L}(F, G)$ の元である 2 つの有界線形写像を合成し $\mathcal{L}(E, G)$ の元である有界線形写像を作り出す写像 B

$$B : (x, y) \in \mathcal{L}(E, F) \times \mathcal{L}(F, G) \mapsto y \circ x \in \mathcal{L}(E, G) \tag{11}$$

を導入する．これは以下の命題による．

Listing 14. NDIFF13:38

```
theorem :: NDIFF13:38
  for E,F,G be RealNormSpace
  ex B be Lipschitzian BilinearOperator of
              R_NormSpace_of_BoundedLinearOperators(E,F),
              R_NormSpace_of_BoundedLinearOperators(F,G),
              R_NormSpace_of_BoundedLinearOperators(E,G)
  st
    for u be Point of R_NormSpace_of_BoundedLinearOperators(E,F),
        v be Point of R_NormSpace_of_BoundedLinearOperators(F,G)
    holds B.(u,v) = v*u;
```

この B と前項で述べた命題 NDIFF13:29 による写像

$$(v' \circ u) \odot u' : \ x \in E \mapsto ((v' \circ u)(x), u'(x)) \in \mathcal{L}(E, F) \times \mathcal{L}(F, G)$$

を用いれば (10) 式は

$$(v \circ u)'(x) = (B \circ ((v' \circ u) \odot u')(x) \tag{12}$$

と書ける．これに前項の命題 NDIFF13:42 を適用すれば，上式の右辺の関数は n 階連続微分可能であり $(v \circ u)'$ は n 階連続微分可能である．

最後に前項同様に命題 NDIFF13:33 を適用して，$v \circ u$ が $n+1$ 階連続微分可能であることが示される．以上で微分の階数に関する帰納法の証明が完成する．

3 陰関数定理の帰納法による証明

微分の階数 n についての帰納法を用いる [1]．$n = 1$ の場合は 1 章で述べた命題 NDIFF9:21 で証明済みである．

微分の階数 n について定理が成立しているとする．ここで，

$$f : (x, y) \in Z \subseteq E \times F \mapsto f(x, y) \in G$$

を Z から G への写像で Z 上で $n+1$ 階連続微分可能とする．また，g を A から B への写像で A に含まれる E の任意の点 $x \in A$ について

$$f(x, g(x)) = c$$

かつ

$$\frac{\partial f}{\partial y}(x, g(x))$$

が可逆であるとする．f は Z 上で $n+1$ 階連続微分可能であるから，以下の命題 NDIFF13:37 により f の微分は Z 上で n 階連続微分可能である．

Listing 15. NDIFF13:37

```
theorem :: NDIFF13:37
  for S,E be RealNormSpace, Z be Subset of S, u be PartFunc of S,E, i be Nat
   st u is_differentiable_on i+1,Z & diff(u,i+1,Z) is_continuous_on Z
  holds
     u`|Z is_differentiable_on i,Z & diff(u`|Z,i,Z) is_continuous_on Z;
```

したがって，帰納法の仮定から g は n 階連続微分可能であり，A に含まれる E の任意の点 $x \in A$ について

$$g'(x) = \left(\frac{\partial f}{\partial y}(x, g(x))\right)^{-1} \circ \frac{\partial f}{\partial x}(x, g(x)) \tag{13}$$

が成り立っている．NDIFF13:65 では

```
for x be Point of E, z be Point of [:E,F:]
  st x in A & z = [x,g.x]
holds diff(g,x) = - (Inv partdiff`2(f,z)) * partdiff`1(f,z);
```

と記述されている．

ノルム空間 E から E 自身への恒等写像

$$id_E : \ x \in E \mapsto x \in E$$

と写像 g を用いて，(4) 式の写像によりノルム空間 E から E と F の直積空間 $E \times F$ への写像

$$id_E \odot g : \ x \in E \mapsto (x, g(x)) \in E \times F \tag{14}$$

を導入する．

次に以下の命題

Listing 16. NDIFF13:66

```
theorem :: NDIFF13:66
  for F,G be non trivial RealBanachSpace
  ex I be PartFunc of
```

```
  R_NormSpace_of_BoundedLinearOperators(F,G),
  R_NormSpace_of_BoundedLinearOperators(G,F)
st dom I = InvertibleOperators(F,G)
 & rng I = InvertibleOperators(G,F)
 & I is one-to-one
 & I is_continuous_on InvertibleOperators(F,G)
 & ( ex J be PartFunc of
       R_NormSpace_of_BoundedLinearOperators(G,F),
       R_NormSpace_of_BoundedLinearOperators(F,G)
     st J = I " & J is one-to-one & dom J = InvertibleOperators(G,F)
      & rng J = InvertibleOperators(F,G) & J is_continuous_on InvertibleOperators(G,F) )
 & ( for u be Point of R_NormSpace_of_BoundedLinearOperators(F,G)
      st u in InvertibleOperators(F,G) holds I.u = Inv u )
 & for n be Nat holds
     I is_differentiable_on n+1,InvertibleOperators(F,G)
   & diff(I,n+1,InvertibleOperators(F,G)) is_continuous_on InvertibleOperators(F,G);
```

を用いて，ノルム空間 E から F への有界線形写像が作る空間 $\mathcal{L}(E,F)$ の元のうち可逆な元の集合を $\mathcal{H}(E,F)$，同様に F から E への可逆な有界線形写像全部の集合を $\mathcal{H}(F,E)$ とし $\mathcal{H}(E,F)$ の元 u を $\mathcal{H}(F,E)$ の元 u^{-1} を対応させる写像 I

$$I:\ u \in \mathcal{H}(E,F) \mapsto u^{-1} \in \mathcal{H}(F,E) \tag{15}$$

を導入する．この写像 I は任意の階数で連続微分可能である．

このの I を用いると，

$$g'(x) = \left(\left(I \circ \frac{\partial f}{\partial y} \circ (id_E \odot g)\right)(x)\right) \circ \left(\left(\frac{\partial f}{\partial x} \circ (id_E \odot g)\right)(x)\right) \tag{16}$$

と書ける．ここで，

$$\left(\frac{\partial f}{\partial x} \circ (id_E \odot g)\right)(x) \in \mathcal{L}(E,G) \tag{17}$$

であり，

$$\left(\frac{\partial f}{\partial y} \circ (id_E \odot g)\right)(x) \in \mathcal{L}(F,G)$$

であるから，

$$\left(I \circ \frac{\partial f}{\partial y} \circ (id_E \odot g)\right)(x) \in \mathcal{L}(G,F) \tag{18}$$

である．

さらに有界連続双線形写像

$$B : (h,k) \in \mathcal{L}(E,G) \times \mathcal{L}(G,F) \mapsto k \circ h \in \mathcal{L}(E,F) \tag{19}$$

を用いると (16),(17),(18) 式から

$$g'(x) = B\left(\left(I \circ \frac{\partial f}{\partial y} \circ (id_E \odot g)\right)(x), \left(\frac{\partial f}{\partial x} \circ (id_E \odot g)\right)(x)\right) \tag{20}$$

を得る．

ノルム空間 S, E, F, G と任意の自然数 n について，$E \times F$ から G への有界連続双線形写像

$$B : (x,y) \in E \times F \mapsto B(x,y) \in G$$

および，n 階連続微分可能写像であるようなそれぞれ S から E への写像 u と，S から F への写像 v が与えられたとき，これらの写像と S から (4) 式により $E \times F$ への写像

$$u \odot v : \ x \in S \mapsto (u(x), v(x)) \in E \times F$$

を用いて合成される S から F への写像

$$B \circ (u \odot v) : \quad x \in S \mapsto (B \circ (u \odot v))(x) = B(u(x), v(x)) \in F$$

は前章で述べたように NDIFF13:42 により n 階連続微分可能である．

ここで，(20) を (4) で定義される写像の合成を用いて書き直すと，

$$g'(x) = \left(B \circ \left(\left(I \circ \frac{\partial f}{\partial y} \circ (id_E \odot g) \right) \odot \left(\frac{\partial f}{\partial x} \circ (id_E \odot g) \right) \right) \right)(x) \tag{21}$$

となる．

$$I \circ \frac{\partial f}{\partial y} \circ (id_E \odot g) \tag{22}$$

は，以下の順に n 階連続微分可能であることが確認できる．

まず，(14) 式で定義される関数 Z

$$id_E \odot g : \ x \in E \mapsto (id \odot g)(x) = (x, g(x)) \in E \times F$$

は，恒等写像 id_E が任意の階数，したがって n 階連続微分可能であり，g が n 階連続微分可能であるから $id_E \odot g$ も同様に n 階連続微分可能である．

これは以下の命題による．

Listing 17. NDIFF13:61

```
theorem :: NDIFF13:61
  for S,E,F be RealNormSpace, u be PartFunc of S,E,
      v be PartFunc of S,F, w be PartFunc of S,[:E,F:],
      Z be Subset of S, i be Nat
   st w = <:u,v:>
    & u is_differentiable_on i+1,Z & diff(u,i+1,Z) is_continuous_on Z
    & v is_differentiable_on i+1,Z & diff(v,i+1,Z) is_continuous_on Z
  holds
    w is_differentiable_on i+1,Z & diff(w,i+1,Z) is_continuous_on Z;
```

次に f は $n+1$ 階連続微分可能であるから，以下の命題 NDIFF13:52 により $\partial f/\partial y$ は n 階連続微分可能である．

Listing 18. NDIFF13:52

```
theorem :: NDIFF13:52
  for i be Nat, E,F,G be RealNormSpace,
      Z be non empty Subset of [:E,F:],
      f be PartFunc of [:E,F:],G
   st f is_differentiable_on i+1,Z
  holds
    f `partial`1| Z is_differentiable_on i,Z & f `partial`2| Z is_differentiable_on i,Z;
```

さらに (15) 式で定義される関数 I

$$I : u \in \mathcal{H}(E, F) \mapsto u^{-1} \in \mathcal{H}(F, E)$$

は任意の階数で，したがって n 階連続微分可能である．これらの 3 つの写像は何れも n 階連続微分可能であるから，前章で述べた 2 つの連続微分可能な写像に合成写像に関する命題 NDIFF13:44 を用いれば (22) 式の写像

$$I \circ \frac{\partial f}{\partial y} \circ (id_E \odot g)$$

が n 階連続微分可能であることが示される．同様の手順で

$$\frac{\partial f}{\partial x} \circ (id_E \odot g) \tag{23}$$

についても f が $n+1$ 階連続微分可能であるから命題 NDIFF13:52 により $\partial f/\partial x$ は n 階連続微分可能であり，これと写像 $id_E \odot g$ が n 階連続微分可能であることから命題 NDIFF13:44 を用いて n 階連続微分可能であることが示される．

以上により，命題 NDIFF13:42 を用いれば　(21) 式

$$g'(x) = \left(B \circ \left(\left(I \circ \frac{\partial f}{\partial y} \circ (id_E \odot g)\right) \odot \left(\frac{\partial f}{\partial x} \circ (id_E \odot g)\right)\right)\right)(x)$$

の右辺の関数が $n+1$ 階連続微分可能であること，したがって g' が n 階連続微分可能であることが示される．

最後に前章で述べた NDIFF13:33 を適用して，g が $n+1$ 階連続微分可能であることが示される．以上で微分の階数に関する帰納法の証明が完成する．

4　帰納法による逆関数定理の証明

前章の陰関数定理 [1] の証明と同様の議論を繰り返すが，冗長性を厭わずに記述すれば以下のようになる．

$n = 0$ の場合は命題 NDIFF10:17 で証明済みである．

微分の階数についての帰納法の仮定により n については定理が成り立つものとする．

E, F をバナハ空間 (RealBanachSpace)，Z を E の開部分集合，

$$f : x \in Z \subseteq E \mapsto f(x) \in F$$

を Z から G への写像で Z 上で $n+1$ 階連続微分可能とする．a, b をそれぞれ E, F の元とし，a は Z に含まれ

$$f(a) = b \tag{24}$$

$$f'(a) \text{ が可逆} \tag{25}$$

とする．

このとき，f は命題 NDIFF13:37 により Z 上で n 階連続微分可能であるから，f の定義域 Z に含まれる a の開近傍 A と b の開近傍 B 及び B から A の上への写像 g が存在し，

$$f(A) = B,\ g(b) = a$$

を満たす．f の A 上への制限 $f|A$ と g は全単射であり，互いに逆写像になっている．しかも，$f|A$ と g はそれぞれ A,B 上 n 階連続微分可能である．

また B の任意の点 $y \in B$ について

$$f'(g(y))$$

は可逆であり，

$$g'(y) = f'(g(y))^{-1} \tag{26}$$

ここで前章で導入した命題 NDIFF13:66 による写像，ノルム空間 E から F への有界線形写像が作る空間 $\mathcal{L}(E,F)$ の元のうち可逆な元の集合を $\mathcal{H}(E,F)$，同様に F から E への可逆な有界線形写像全部の集合を $\mathcal{H}(F,E)$ とし $\mathcal{H}(E,F)$ の元 u を $\mathcal{H}(F,E)$ の元 u^{-1} を対応させる写像 I を用いる．

$$I : u \in \mathcal{H}(E,F) \mapsto u^{-1} \in \mathcal{H}(F,E)(\text{再出}) \tag{27}$$

前章で述べたようにこの写像 I は任意の階数で，したがって n 階連続微分可能である．

この写像 I を用いて (26) 式を書き直すと

$$g'(y) = I(f'(g(y))) = (I \circ f' \circ g)(y) \tag{28}$$

を得る．上式の右辺の写像

$$I \circ f' \circ g \tag{29}$$

について前章と同様の議論を行う．上式中の 3 つの写像

$$I,\ f',\ g$$

は何れも n 階連続微分可能であるから，前章で用いた 2 つの連続微分可能な写像に合成写像に関する命題 NDIFF13:44 を用いれば (29) 式の写像が n 階連続微分可能であることが示される．したがって (28) 式により g' も n 階連続微分可能である．最後に前章同様に NDIFF13:33 を適用して g が $n+1$ 階連続微分可能であることが示され，微分の階数に関する帰納法の証明が完成する．

5 まとめ

ノルム空間上の高階微分の陰関数定理及び逆関数定理とそれらの証明の形式化について報告した．筆者らは既にこれ以外に基づいて，ノルム空間上の Taylor 展開や変分法 [1] などの定理と証明を形式化した．さらに，それらの諸定理の準備の上に微分多様体の形式化を開始している．これらについては次回以降に報告する．

参考文献

[1] Schwartz L. Cours d'analyse. Hermann; 1981.

[2] Nakasho K, Shidama Y. Implicit Function Theorem. Part II. Formalized Mathematics. 2019;27(**2**):117–131.

[3] Nakasho K, Futa Y. Inverse Function Theorem. Part I. Formalized Mathematics. 2021;29(**1**):9–19.

[4] Nakasho K, Shidama Y. On Implicit and Inverse Function Theorems on Euclidean Spaces. Formalized Mathematics. 2022;30(**1**):165–175.

[5] Shidama Y, Nakasho K. On The Formalizations for Higher Derivatives of Vector Valued Functions. Mechanized Mathematics and Its Applications, Works in Progress (MMA-WiP). 2022;4(2):1–16. Available from: `https://mizar-jp.org/journal/MMA-WiP/j2022.4.2`.

Mizar article information

Works in Progress

NDIFF12 *Derivatives of Lipschitzian Bilinear Operators*
by Kazuhisa Nakasho and Yasunari Shidama
Summary: In this article, we formalized some fundamentally important theorems for higher-order derivatives of bounded bilinear operators on normed spaces.

NDIFF13 *Differentiation of composite functions*
by Kazuhisa Nakasho and Yasunari Shidama
Summary: In this article, we formalized some fundamentally important theorems for higher-order derivatives of composite functions on norm spaces. Using these theorems, we proved the inverse function theorem for norm-valued functions of normed spaces.